Určitý integrál

Mathematica spočte hodně určitých integrálů, ale musíme být opatrní a ne všem výsledkům věřit. Víme, že je více druhů určitých integrálů (Newtonův, Riemannův, Lebesgueův, Kurzweilův,...), ale z výsledku nepoznáme, o jaký integrál se jedná. Většinou se jedná o zobecněný Newtonův integrál (i když v příručce k programu *Mathematica* se píše hlavně o Riemannově integrálu), jsou však i vyjímky.

Bezproblémové integrály

V případech, že program zná zobecněnou primitivní funkci (viz předchozí kapitolu), integraci program provede (někdy ale pomocí funkcí, které neznáme).

```
Integrate[Sin[x], {x, 0, 3 Pi}]
```

2

```
Integrate[Log[x], {x, 0, 1}]
```

-1

```
Integrate[(Cos[x])^3 , {x, 0, Pi / 2}]
```

$\frac{2}{3}$

```
Integrate[ArcTan[x], {x, 0, Infinity}]
```

— *Integrate::idiv : Integral of* $\tan^{-1}(x)$ *does not converge on* $\{0, \infty\}$. ≫

$\int_0^{\infty} \tan^{-1}(x)\, dx$

```
Integrate[Sin[x], {x, 0, Infinity}]
```

— *Integrate::idiv : Integral of* $\sin(x)$ *does not converge on* $\{0, \infty\}$. ≫

$\int_0^{\infty} \sin(x)\, dx$

```
Integrate[1 / (1 + 3 (Cos[x]) ^2), {x, 0, Pi}]
```

$$\frac{\pi}{2}$$

```
Integrate[Sqrt[Cos[x] - (Cos[x]) ^3], {x, -Pi / 2, Pi / 2}]
```

$$\frac{4}{3}$$

```
Integrate[1 / (x + Sqrt[x^2 + x + 1]), {x, 0, 1}]
```

$$-2+\sqrt{3}+\frac{1}{4}\log\left(432\left(7-4\sqrt{3}\right)\right)$$

```
Integrate[1 / (x^4 + 1), {x, 0, 2}]
```

$$\frac{2\pi+\log\left(\frac{1}{17}\left(33+20\sqrt{2}\right)\right)-2\tan^{-1}\left(\frac{2\sqrt{2}}{3}\right)}{4\sqrt{2}}$$

Program vypočte i integrály funkcí, s jejichž primitivními funkcemi měl problémy, např. signum nebo absolutní hodnota, nebo kdy výsledkem byla nespojitá primitivní funkce.

```
Integrate[Abs[x], {x, -2, -1}]
```

$$\frac{3}{2}$$

```
Integrate[Sign[x], {x, -3, 1}]
```

$$-2$$

```
Integrate[1 / (2 + Sin[x]), {x, 0, Pi}]
```

$$\frac{2\pi}{3\sqrt{3}}$$

```
Integrate[1 / Sqrt[Abs[x]], {x, -1, 1}]
```

4

Následující integrovaná funkce nemá (pro nás) známou primitivní funkci, nicméně *Mathematica* zná tuto primitivní funkci pod názvem Si a dovede tedy spočítat i určité integrály této funkce (ten prvý ale musí spočítat každý).

```
Integrate[Sin[x] / Abs[x], {x, -2, 2}]
```

0

```
Integrate[Sin[x] / Abs[x], {x, -1, 2}]
```

$\mathrm{Si}(2) - \mathrm{Si}(1)$

Mírně problematické výsledky

Výsledky určitých integrálů bývají, na rozdíl od neurčitých integrálů, udány s podmínkou, kdy výsledek platí, ale není tomu tak vždy. Následující výsledek dá spravnou podmínku, kdy integrál konverguje, ale hned další jednodušší příklad ve výsledku zapomene na podmínku n≠-1.

```
Integrate[x^n, {x, 0, 2}]
```

$$\mathrm{ConditionalExpression}\left[\frac{2^{n+1}}{n+1}, \mathrm{Re}(n) > -1\right]$$

```
Integrate[x^n, {x, 1, 2}]
```

$$\frac{2^{n+1} - 1}{n+1}$$

Mathematica klidně zintegruje funkci i mimo její definiční obor. Je to opět proto, že pracuje s komplexními čísly, jak je vidět z následujících příkladů. Program použije automaticky rozdíl hodnot primitivní funkce v krajních bodech - ale někdy se jedná o víceznačné funkce a není proto dobré výsledkům věřit.

```
Integrate[ArcSin[x], {x, 0, 2}]
```

$-1 + i\sqrt{3} + 2\sin^{-1}(2)$

```
Integrate[Sqrt[x], {x, -1, 2}]
```

$\frac{2}{3}\left(2\sqrt{2} + i\right)$

Další příklad je spočten dobře, ale nedostaneme očekávaný výsledek n!, i kfyž dáme předpoklad, že n je celé číslo - ve výsledku je stále Gama funkce.

```
Integrate[x^n Exp[-x], {x, 0, Infinity}, Element[n, Integers]]
```

$\text{ConditionalExpression}[\Gamma(n+1)\,(n \in \mathbb{Z}),\ \text{Re}(n) > -1]$

S některými pro nás snadnými integrály má *Mathematica* časové problémy a jejich výpočet trvá značně déle než u podobných příkladů.

```
Integrate[Sqrt[a^2 - x^2], {x, -a, a}] // Timing
```

$\left\{6.224,\ \frac{1}{2}\pi a\sqrt{a^2}\right\}$

Musíme si být vědomi, že ve výsledcích mohou být pro nás neznámé funkce nebo konstanty (v našich případech se jedná o Besselovu funkci a Eulerovu konstantu).

```
Integrate[Cos[Sin[x]], {x, 0, 2 Pi}]
```

$2\pi J_0(1)$

```
Integrate[Log[x] Exp[-x^2], {x, 0, Infinity}]
```

$-\frac{1}{4}\sqrt{\pi}\,(\gamma + \log(4))$

Více problematické integrály

U následující funkce program primitivní funkci nezná a integrál nespočte ani po čekaní několika minut. My víme bez počítání, že výsledek je roven 0, protože integrujeme spojitou lichou funkci (po dodefinování v 0).

```
Integrate[Sin[x] / Log[Abs[x]], {x, -1 / 2, 1 / 2}]
```

$Aborted

Další funkce má Newtonův integrál, ale nemá ani nevlastní Riemannův ani Lebesgueův integrál. Přesto ho *Mathematica* spočte pomocí primitivní funkce.Druhý příklad nespočte ani po několika minutách - my víme, že výsledkem je +∞.

```
Integrate[Sin[x] / x, {x, 0, Infinity}]
```

$\frac{\pi}{2}$

```
Integrate[Abs[Sin[x]] / x, {x, 0, Infinity}]
```

$Aborted

Ani další příklad program nespočte.Jedná se sice o zobecněný Newtonův integrál, ale se spočetnou množinou bodů nespojitosti integrované funkce. My však jeho hodnotu určíme snadno.Hodnotou je součet nekonečné řady $\Sigma\ 1/\ k^2$, což je $\pi^2/6$. Nelze použít ani limitní proces, pouze numerický výpočet, který se ale od správného výsledku dost liší (ve výpočtu se zřejmě akumulovaly chyby).

```
f[x_] := 1 / (Floor[x]) ^2;
Plot[f[x], {x, 1., 9}, PlotRange → {0, 1}, AxesOrigin → {0, 0}]
```




```
Integrate[f[x], {x, 1, Infinity}]
```

$$\int_1^{\infty} \frac{1}{\lfloor x \rfloor^2}\, dx$$

```
Limit[Integrate[f[x], {x, 1, n}] // N, n → Infinity]
```

— *NIntegrate::nlim : x = n is not a valid limit of integration.* ≫

— *NIntegrate::nlim : x = n is not a valid limit of integration.* ≫

— *NIntegrate::nlim : x = n is not a valid limit of integration.* ≫

— *General::stop :*

Further output of NIntegrate::nlim will be suppressed during this calculation. ≫

$$\lim_{n\to\infty} \text{NIntegrate}\left[\frac{1}{\lfloor x \rfloor^2}, \{x, 1, n\}\right]$$

```
NIntegrate[f[x], {x, 1, Infinity}, WorkingPrecision → 100]
```

— *NIntegrate::slwcon :*
Numerical integration converging too slowly; suspect one of the following: singularity, value of the integration is 0, highly oscillatory integrand, or WorkingPrecision too small. ≫

— *NIntegrate::ncvb : NIntegrate failed to converge to prescribed accuracy after 9 recursive bisections in x near {x} =*
{2.99415578964994043422880129869569039235852172710655767206065 ⋱ 45300996538267682694682570766317257497 67}.
NIntegrate obtained
1.64456241380256522060034087577758868919231410775461964428789 7 ⋱ 3605923473475763931320221525259942418 41 and
0.00018435926854840246628863343298609949481823367311964014790 7 ⋱ 32556018449665966585083681165032442660 31176
for the integral and error estimates. ≫

```
1.644562413802565220600340875777588689192314107754619644287897360592347⋱
  3475763931320221525259942418 41
```

```
Pi^2 / 6 - %
```

```
0.00037165304566121587207429086843650002663579345217879344766086877766 0⋱
  12282680774181147637462551686 5
```

Příklad existence Newtonova a nikoli Riemannova integrálu

Stačí vzít vhodnou derivaci nějaké funkce, která bude hodně oscilovat.Program vypočte integrál jako rozdíl hodnot primitivní funkce (tedy Newtonův integrál) - Riemannův integrálů neexistuje, což naznačuje poslední graf kmitajících částečných součtů.

```
Clear[g]; g[x_] := D[x^2 * Cos[1 / x^2], x];
GraphicsRow[
 {Plot[Evaluate[g[x]], {x, .1, 2}, AxesOrigin → {0, 0},
   PlotLabel → "g(x)"],
  Plot[x^2 * Cos[1 / x^2], {x, .1, 2}, AxesOrigin → {0, 0},
   PlotLabel → "∫g(x) ⅆx", ImageSize → 260] }]
```




```
Integrate[g[x], {x, 0, 1}]
```

cos(1)

```
Clear[h];
h[n_] :=
 1 / n * Sum[2 * k / n * Cos[n^2 / k^2] + 2 * n / k * Sin[n^2 / k^2],
    {k, 1, n}] // N
```

```
Manipulate[Plot[h[Floor[x]], {x, 1, n},
  PlotRange → {{0, 300}, {-4, 4}}], {n, 100, 300, 10}]
```



Konvergence integrálů

Pokud program dovede integrál s parametrem spočítat, napíše většinou do výsledku i podmínky na parametr, kdy výsledek platí. Někdy však napíše jen čás podmínek (příklad s kvadratickým trojčlenem). Pokud si s integrálem neví rady, samozřejmě žádně podmínky na konvergenci nedostaneme.

```
Integrate[(1 - Cos[x]) / x^a, {x, 0, Pi}]
```

$$\text{ConditionalExpression}\left[\frac{\pi^{1-a}\left({}_1F_2\left(\frac{1}{2}-\frac{a}{2};\frac{1}{2},\frac{3}{2}-\frac{a}{2};-\frac{\pi^2}{4}\right)-1\right)}{a-1},\ \text{Re}(a)<3\right]$$

```
Integrate[Sin[x] / x^a, {x, 0, Infinity}]
```

$$\text{ConditionalExpression}\left[\cos\left(\frac{\pi a}{2}\right)\Gamma(1-a),\ 0<\text{Re}(a)<2\right]$$

```
Integrate[1 / (x^2 + a x + 1), {x, 0, 1}]
```

$$\text{ConditionalExpression}\left[-\frac{2\left(\tan^{-1}\left(\frac{a}{\sqrt{4-a^2}}\right)-\tan^{-1}\left(\frac{a+2}{\sqrt{4-a^2}}\right)\right)}{\sqrt{4-a^2}},\right.$$
$$\left(\sqrt{a^2-4}+a \notin \mathbb{R} \bigvee \operatorname{Re}\left(\sqrt{a^2-4}+a\right) \geq 0 \bigvee \operatorname{Re}\left(\sqrt{a^2-4}+a\right) \leq -2\right) \bigwedge$$
$$\left.\left(\operatorname{Re}(a) \geq \operatorname{Re}\left(\sqrt{a^2-4}\right) \bigvee \operatorname{Re}(a)+2 \leq \operatorname{Re}\left(\sqrt{a^2-4}\right) \bigvee \sqrt{a^2-4}-a \notin \mathbb{R}\right)\right]$$

```
Integrate[x^a (ArcTan[x])^b, {x, 0, Infinity}]
```

$$\int_0^\infty x^a \tan^{-1}(x)^b \, dx$$